
You are looking at the README for the main branch. The README for the latest stable release is located here.

Temple
Temple is an Elixir DSL for writing HTML and SVG.

  
    
  
  Installation


Add temple to your list of dependencies in mix.exs:
def deps do
  [
    {:temple, "~> 0.12"}
  ]
end

  
    
  
  Goals


Currently Temple has the following things on which it won't compromise.
	Will only work with valid Elixir syntax.
	Should work in all web environments such as Plug, Aino, Phoenix, and Phoenix LiveView.


  
    
  
  Usage


Using Temple is as simple as using the DSL inside of an temple/1 block. The runtime result of the macro is your HTML.
See the guides for more details.
import Temple

temple do
  h2 do: "todos"

  ul class: "list" do
    for item <- @items do
      li class: "item" do
        div class: "checkbox" do
          div class: "bullet hidden"
        end

        div do: item
      end
    end
  end

  script do: """
  function toggleCheck({currentTarget}) {
    currentTarget.children[0].children[0].classList.toggle("hidden");
  }

  let items = document.querySelectorAll("li");

  Array.from(items).forEach(checkbox => checkbox.addEventListener("click", toggleCheck));
  """
end

  
    
  
  Components


Temple components are simple to write and easy to use.
Unlike normal partials, Temple components have the concept of "slots", which are similar Vue. You can also refer to HEEx and Surface for examples of templates that have the "slot" concept.
Temple components are compatible with HEEx and Surface components and can be shared.
Please see the guides for more details.
defmodule MyAppWeb.Component do
  import Temple

  def card(assigns) do
    temple do
      section do
        div do
          slot @header
        end

        div do
          slot @inner_block
        end

        div do
          slot @footer
        end
      end
    end
  end
end
Using components is as simple as passing a reference to your component function to the c keyword.
import MyAppWeb.Component

c &card/1 do
  slot :header do
    @user.full_name
  end

  @user.bio

  slot :footer do
    a href: "https://twitter.com/#{@user.twitter}" do
      "@#{@user.twitter}"
    end
    a href: "https://github.com/#{@user.github}" do
      "@#{@user.github}"
    end
  end
end

  
    
  
  Engine


By default, Temple will use the EEx.SmartEngine that is built into the Elixir standard library. If you are a web framework that uses it's own template engine (such as Aino and Phoenix/LiveView, you can configure Temple to it!
# config/config.exs

config :temple,
  engine: Aino.View.Engine # or Phoenix.HTML.Engine or Phoenix.LiveView.Engine

  
    
  
  Formatter


To include Temple's formatter configuration, add :temple to your .formatter.exs.
[
  import_deps: [:temple],
  inputs: ["*.{ex,exs}", "priv/*/seeds.exs", "{config,lib,test}/**/*.{ex,exs,lexs}"],
]

  
    
  
  Phoenix


When using Phoenix ~> 1.7, all you need to do is include :temple in your mix.exs.
If you plan on using the template structure that < 1.6 Phoenix applications use, you can use :temple_phoenix as described below.
To use with Phoenix, please use the temple_phoenix package! This bundles up some useful helpers as well as the Phoenix Template engine.

  
    
  
  Related


	Introducing Temple: An elegant HTML library for Elixir and Phoenix
	Temple, AST, and Protocols
	Thinking Elixir Episode 92: Temple with Mitchell Hanberg
	How EEx Turns Your Template Into HTML



  

    
Getting Started
    


  
    
  
  Install


Welcome!
Temple is a HTML DSL for Elixir, let's get started!
First, make sure you are using Elixir V1.13 or higher.
Add :temple to your deps and run mix deps.get
{:temple, "~> 0.9.0-rc.0"}
Now you must prepend the Temple compiler to your projects :compilers configuration in mix.exs. There is a chance that your project doesn't set this option at all, but don't worry, it's really easy to add!
defmodule MyApp.MixProject do
  use Mix.Project

  def project do
    [
      # ...
      compilers: [:temple] ++ Mix.compilers(),
      # ...
    ]
  end

# ...

end
All done, Now let's start building our app!

  
    
  
  Configuration


Temple works out of the box without any configuration, but here are a couple of conifg options that you could need to use.

  
    
  
  Engine


By default, Temple uses the built in Phoenix.HTML.Engine. If you want to use a different engine, this is as easy as setting the :engine configuration option.
You can also configure the function that is used for runtime attributes. By default, Temple uses Phoenix.HTML.attributes_escape/1.
# config/config.exs

config :temple,
  engine: EEx.SmartEngine,
  attributes: {Temple, :attributes}

  
    
  
  Aliases


Temple code will reserve some local function calls for HTML tags. If you have a local function that you would like to use instead, you can create an alias for any tag.
Common aliases for Phoenix projects look like this:
config :temple,
  aliases: [
    label: :label_tag,
    link: :link_tag,
    select: :select_tag,
    textarea: :textarea_tag
  ]


  

    
Your First Template
    

A Temple template is written inside of the Temple.temple/1 macro. Code inside there will be compiled into efficient Elixir code by the configured EEx engine. 
Local functions that have a corresponding HTML5 tag are reserved and will be used when generated your markup. Let's take a look at a basic form written with Temple.
defmodule MyApp.FormExample do
  import Temple

  def form_page() do
    assigns = %{title: "My Site | Sign Up", logged_in: false}

    temple do
      "<!DOCTYPE html>"

      html do
        head do
          meta charset: "utf-8"
          meta http_equiv: "X-UA-Compatible", content: "IE=edge"
          meta name: "viewport", content: "width=device-width, initial-scale=1.0"
          link rel: "stylesheet", href: "/css/app.css"

          title do: @title
        end

        body do
          if @logged_in do
            header class: "header" do
              ul do
                li do
                  a href: "/", do: "Home"
                end
                li do
                  a href: "/logout", do: "Logout"
                end
              end
            end
          end

          form action: "", method: "get", class: "form-example" do
            div class: "form-example" do
              label for: "name", do: "Enter your name:"
              input type: "text", name: "name", id: "name", required: true
            end
            div class: "form-example" do
              label for: "email", do: "Enter your email:"
              input type: "email", name: "email", id: "email", required: true
            end
            div class: "form-example" do
              input type: "submit", value: "Subscribe!"
            end
          end
        end
      end
    end
  end
end
This example showcases an entire HTML page made with Temple! Let's dive a little deeper everything we're seeing here.
Through out this guide, you will see code that includes features that are explained later on. Feel free to skip ahead to read on, or just keep reading. It will all make sense eventually!

  
    
  
  Text Nodes


The text node is a basic building block of any HTML document. In Temple, text nodes are represented by Elixir string literals.
The very first line of the previous example is our doc type, emitted into the final document with "<!DOCTYPE html>". This is a text node and will be emitted into the document as-is.
Note: String literals are emitted into text nodes. If you are using string interpolation with the #{some_expression} syntax, that is treated as an expression and will be evaluated in whichever way the configured engine evaluates expression. By default, the EEx.SmartEngine doesn't do any escaping of expressions, so that could still be emitted as-is, or even as HTML to be interpreted by your web browser.

  
    
  
  Void Tags


Void tags are HTML5 tags that do not have children, meaning they are "self closing".
We can observe these in the previous example as the <input> tag. You'll note that the tag does not have a :do key or a do block.

  
    
  
  Non-void Tags


Non-void tags are HTML5 tags that do have children. You are probably most familiar with these type of tags, as they include the famous <div></div> and <span></span>.
These tags can enclose their children nodes with either a do/end block or the inline :do keyword.

  
    
  
  Whitespace


Nonvoid tags that use the do/end syntax will be emitted with internal whitespace.
temple do
  div class: "foo" do
    # children
  end
end
...will emit markup that looks like...
<div class="foo">
  <!-- children -->
</div>
Note: The Elixir comment will not be rendered into an HTML comment. This is just used in the example. (This does sound like a good feature though...)
Nonvoid tags that use the :do keyword syntax will be emitted without internal whitespace. This allows you to correctly use the :empty CSS psuedo-selector in your stylesheet. 
temple do
  p class: "alert alert-info", do: "Your account was recently updated!"
end
...will emit markup that looks like...
<p class="alert alert-info">Your account was recently updated!</p>

  
    
  
  Attributes


Temple leverages Phoenix.HTML.attributes_escape/1 internally, so you can refer to it's documentation for all of the details.

  
    
  
  Dynamic Attributes


To render dynamic attributes into a tag, you can pass them with the reserved attribute :rest!.
assigns = %{
  data: [data_foo: "hi"]
}

temple do
  div id: "foo", rest!: @data do
    "Hello, world!"
  end
end
will render to
<div id="foo" data-foo="hi">
  Hello, world!
</div>

  
    
  
  Elixir Expressions


Any Elixir expression can be used anywhere inside of a Temple template. Here are a few examples.
temple do
  h2 do: "Members"

  ul do
    for member <- @members do
      li do: member
    end
  end
end

  
    
  
  Match Expressions


Match expression are handled slightly differently. Generally if you are assigning an expression to a variable (a match), you are going to use that binding later and do not want to emit it into the document.
So, match expressions are not emitted into the document. They are functionally equivalent to the <% .. %. syntax of EEx. The expression is evaluated, but not included in the rendered document.
Typically you should not be writing this type of expression inside of your template, but if you wanted to declare an alias, you would need to write the following to not emit the alias into the document.
temple do
  _ = alias My.Deep.Module

  div do
    Module.func()
  end
end

  
    
  
  Assigns


Since Temple uses the EEx.SmartEngine by default, you are able to use the assigns feature.
The assigns feature allows you to ergonomically access the members of a assigns variable by the @ macro.
The assign variable just needs to exist within the scope of the template (the same as a normal EEx template that uses EEx.SmartEngine), it can be a function parameter or created inside the function.
def card(assigns) do
  temple do
    div class: "card" do
      section class: "card-header" do
        @name
      end

      section class: "card-body" do
        @bio
      end

      if Enum.any?(@socials) do
        section class: "card-footer" do
          for social <- @socials do
            a href: social.link do
              social.name
            end
          end
        end
      end
    end
  end
end


  

    
Components
    

Temple has the concept of components, which allow you an expressive and composable way to break up your templates into re-usable chunks.
A component is any arity-1 function that take an argument called assigns and returns the result of the Temple.temple/1 macro.

  
    
  
  Definition


Here is an example of a simple Temple component. You can observe that it seems very similar to a regular Temple template, and that is because it is a regular template!
defmodule MyApp.Components do
  import Temple

  def button(assigns) do
    temple do
      button type: "button", class: "bg-blue-800 text-white rounded #{@class}" do
        @text
      end
    end
  end
end

  
    
  
  Usage


To use a component, you will use the special c keyword. This is called a "keyword" because it is not a function or macro, but only exists inside of the Temple.temple/1 block.
The first argument will be the function reference to your component function, followed by any assigns. You can pass dynamic assigns using the :rest! keyword the same way you would with a normal tag.
defmodule MyApp.ConfirmDialog do
  import Temple
  import MyApp.Components

  def render(assigns) do
    temple do
      dialog open: true do
        p do: "Are you sure?"
        form method: "dialog" do
          c &button/1, class: "border border-white", text: "Yes"
        end
      end
    end
  end
end

  
    
  
  Slots


Temple components can take "slots" as well. This is the method for providing dynamic content from the call site into the component.
Slots are defined and rendered using the slot keyword. This is similar to the c keyword, in that it is not defined using a function or macro.

  
    
  
  Default Slot


The default slot can be rendered from within your component by passing the slot the @inner_block assign. Let's redefine our button component using slots.
defmodule MyApp.Components do
  import Temple

  def button(assigns) do
    temple do
      button type: "button", class: "bg-blue-800 text-white rounded #{@class}" do
        slot @inner_block
      end
    end
  end
end
You can pass content through the "default" slot of your component simply by passing a do/end block to your component at the call site. This is a special case for the default slot.
defmodule MyApp.ConfirmDialog do
  import Temple
  import MyApp.Components

  def render(assigns) do
    temple do
      dialog open: true do
        p do: "Are you sure?"
        form method: "dialog" do
          c &button/1, class: "border border-white" do
            "Yes"
          end
        end
      end
    end
  end
end

  
    
  
  Named Slots


You can also define a "named" slot, which allows you to pass more than one set of dynamic content to your component.
We'll use a "card" example to illustrate this. This example is adapted from the Surface documentation on slots.
Definition
defmodule MyApp.Components do
  import Temple

  def card(assigns) do
    temple do
      div class: "card" do
        header class: "card-header", style: "background-color: @f5f5f5" do
          p class: "card-header-title" do
            slot @header
          end
        end

        div class: "card-content" do
          div class: "content" do
            slot @inner_block
          end
        end

        footer class: "card-footer", style: "background-color: #f5f5f5" do
          slot @footer
        end
      end
    end
  end
end
Usage
def MyApp.CardExample do
  import Temple
  import MyApp.Components

  def render(assigns) do
    temple do
      c &card/1 do
        slot :header do
          "A simple card component"
        end

        "This example demonstrates how to create components with multiple, named slots"

        slot :footer do
          a href: "#", class: "card-footer-item", do: "Footer Item 1"
          a href: "#", class: "card-footer-item", do: "Footer Item 2"
        end
      end
    end
  end
end

  
    
  
  Passing data to and through Slots


Sometimes it is necessary to pass data into a slot (hereby known as slot attributes) from the call site and from a component definition (hereby known as slot arguments) back to the call site. Dynamic slot attributes can be passed using the :rest! attribute in the same way you can with tag attributes.
Let's look at what a table component could look like. Here we observe we access an attribute in the slot in the header with col.label.
This example is taken from the HEEx documentation to demonstrate how you can build the same thing with Temple.
Note: Slot attributes can only be accessed on an individual slot, so if you define a single slot definition, you still need to loop through it to access it, as they are stored as a list.
Definition
defmodule MyApp.Components do
  import Temple

  def table(assigns) do
    temple do
      table do
        thead do
          tr do
            for col <- @col do
              th do: col.label # 👈 accessing a slot attribute
            end
          end
        end

        tbody do
          for row <- @rows do
            tr do
              for col <- @col do
                td do
                  slot col, row
                end
              end
            end
          end
        end
      end
    end
  end
end
Usage
When we render the slot, we can pattern match on the data passed through the slot via the :let attribute.
def MyApp.TableExample do
  import Temple
  import MyApp.Componens

  def render(assigns) do
    temple do
      section do
        h2 do: "Users"

        c &table/1, rows: @users do
          #          👇 defining the parameter for the slot argument
          slot :col, let!: user, label: "Name" do # 👈 passing a slot attribute
            user.name
          end

          slot :col, let!: user, label: "Address" do
            user.address
          end
        end
      end
    end
  end
end


  

    
Converting HTML
    

If you want to use something like TailwindUI with Temple, you're going to have to convert a ton of vanilla HTML into Temple syntax.
Luckily, Temple provides a mix task for converting an HTML file into Temple syntax and writes it to stdout.

  
    
  
  Usage


First, we would want to create a temporary HTML file with the HTML we'd like to convert.
Hint
The following examples use the pbpaste and pbcopy utilities found on macOS. These are used to send your clipboard contents into stdout and put stdout into your clipboard.
$ pbpaste > temp.html

Then, we can convert that file and copy the output into our clipboard.
$ mix temple.convert temp.html | pbcopy

Now, you are free to paste the new temple syntax into your project!


  

    
Migrating from 0.8 to 0.9
    

First off, Temple now requires Elixir 1.13 or higher. This is because of some changes that were brought to the Elixir parser.

  
    
  
  Whitespace Control


To control whitespace in an element, Temple will now control this based on whether the do was used in the keyword list syntax or the do/end syntax.
In 0.8, you would do:
span do
  "hello!"
end

# <span>
#   hello!
# </span>

# The ! version of the element would render it as "tight"
span! do
  "hello!"
end

# <span>hello!</span>
In 0.9, you would do:
span do
  "hello!"
end

# <span>
#   hello!
# </span>

span do: "hello!"

# <span>hello!</span>

  
    
  
  Components


Components are no longer module based. To render a component, you can pass a function reference to the c keyword. You also no longer need to define a component in a module, using the Temple.Component module and its render macro.
In 0.8, you would define a component like:
defmodule MyAppWeb.Component.Card do
  import Temple.Component

  render do
    div class: "border p-4 rounded" do
      slot :default
    end
  end
end
And you would use the component like:
div do
  c MyAppWeb.Component.Card do
    "Welcome to my app!"
  end
end
In 0.9, you would define a component like:
defmodule MyAppWeb.Components do
  import Temple

  def card(assigns) do
    temple do
      div class: "border p-4 rounded" do
        slot :default
      end
    end
  end
end
And you would use the component like:
div do
  c &MyAppWeb.Components.card/1 do
    "Welcome to my app!"
  end
end
We can observe here that in 0.9 the component is just any 1-arity function, so you can define them anywhere and you can have more than 1 in a single module.

  
    
  
  defcomp


Now that components are just functions, you no longer need this special macro to define a component in the middle of the module.
This can simply be converted to a function.

  
    
  
  Phoenix


All Phoenix related items have moved to the temple_phoenix package. Please see that library docs for more details.


  

    
Migrating from 0.10 to 0.11
    

Most of the changes in this release are related to tweaking Temple's component model to align with HEEx & Surface.

  
    
  
  Rendering Slots


Slots are now available as assigns in the component and are rendered as such.

  
    
  
  Before


def my_component(assign) do
  temple do
    span do
      slot :a_slot
    end
  end
end

  
    
  
  After


def my_component(assign) do
  temple do
    span do
      slot @a_slot
    end
  end
end

  
    
  
  :default slot has been renamed to :inner_block


The main body of a component has been renamed from :default to :inner_block.
Note: The "after" example also includes the necessary change specified above.

  
    
  
  Before


def my_component(assign) do
  temple do
    span do
      slot :default
    end
  end
end

  
    
  
  After


def my_component(assign) do
  temple do
    span do
      slot @inner_block
    end
  end
end

  
    
  
  Passing data into slots


The syntax for capturing data being passed from the call site of a slot to the definition of a slot (or put another way, from the definition of a component to the call site of the component) has changed. 
You now capture it as the value of the :let! attribute on the slot definition.

  
    
  
  Before


def my_component(assign) do
  temple do
    c &my_component/1 do
      slot :a_slot, %{some: value} do
        "I'm using some #{value}"
      end
    end
  end
end

  
    
  
  After


def my_component(assign) do
  temple do
    c &my_component/1 do
      slot :a_slot, let!: %{some: value} do
        "I'm using some #{value}"
      end
    end
  end
end


  

    
Temple 
    



      
Temple syntax is available inside the temple, and is compiled into efficient Elixir code at compile time using the configured EEx.Engine.
You should checkout the guides for a more in depth explanation.

  
    
  
  Usage


defmodule MyApp.HomePage do
  import Temple

  def render() do
    assigns = %{title: "My Site | Sign Up", logged_in: false}

    temple do
      "<!DOCTYPE html>"

      html do
        head do
          meta charset: "utf-8"
          meta http_equiv: "X-UA-Compatible", content: "IE=edge"
          meta name: "viewport", content: "width=device-width, initial-scale=1.0"
          link rel: "stylesheet", href: "/css/app.css"

          title do: @title
        end

        body do
          header class: "header" do
            ul do
              li do
                a href: "/", do: "Home"
              end
              li do
                if @logged_in do
                  a href: "/logout", do: "Logout"
                else
                  a href: "/login", do: "Login"
                end
              end
            end
          end

          main do
            "Hi! Welcome to my website."
          end
        end
      end
    end
  end
end

  
    
  
  Configuration



  
    
  
  Engine


By default Temple wil use the EEx.SmartEngine, but you can configure it to use any other engine. Examples could be Phoenix.HTML.Engine or Phoenix.LiveView.Engine.
config :temple, engine: Phoenix.HTML.Engine

  
    
  
  Aliases


You can add an alias for an element if there is a namespace collision with a function. If you are using Phoenix.HTML, there will be namespace collisions with the <link> and <label> elements.
config :temple, :aliases,
  label: :label_tag,
  link: :link_tag,
  select: :select_tag

temple do
  label_tag do
    "Email"
  end

  link_tag href: "/css/site.css"
end
This will result in:
<label>
  Email
</label>

<link href="/css/site.css">

Compiles runtime attributes.
To use this function, you set it in application config.
By default, Temple uses {Phoenix.HTML, :attributes_escape}. This is useful if you want to use EEx.SmartEngine.
config :temple,
  engine: EEx.SmartEngine,
  attributes: {Temple, :attributes}
Note
This function does not do any HTML escaping
Note
This function is used by the compiler and shouldn't need to be used directly.

  



  
    
      
    
    
      temple(block)

Temple.Component 
    



      
Use this module to create your own component implementation.
This is only required if you are not using a component implementation from another framework,
like Phoenix LiveView.
At it's core, a component implmentation includes the following functions
	component/2
	inner_block/2
	render_slot/2

These functions are used by the template compiler, so you won't be calling them directly.

  
    
  
  Usage


Invoke the __using__/1 macro to create your own module, and then import that module where you
need to define define or use components (usually everywhere).
We'll use an example that is similar to what Temple uses in its own test suite..
defmodule MyAppWeb.Component do
  use Temple.Component

  defmacro __using__(_) do
    quote do
      import Temple
      import unquote(__MODULE__)
    end
  end
end
Then you can use your module when you want to define or use a component.
defmodule MyAppWeb.Components do
  use MyAppWeb.Component

  def basic_component(_assigns) do
    temple do
      div do
        "I am a basic component"
      end
    end
  end
end

      




  

    
mix compile.temple 
    






  

    
mix temple.convert 
    



      
This task is useful for converting a ton of HTML into Temple syntax.
Note about EEx and HEEx
In the future, this should be able to convert EEx and HEEx as well, but that would involve invoking or forking their parsers. That is certainly doable, but is out of scope for what I needed right now. Contributions are welcome!

  
    
  
  Usage


$ mix temple.convert some_file.html

